
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 22.6.5 which is part of the NBIOT test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more NBIOT UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_22_6_5
Test Group:
/NBIOT/22_6/NBIOT_ESM_Testcases.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D17wk37 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Spreadtrum RDA8909
Verification Status:
PASS
4 Corrections required for test case 22.6.5
4.1 Introduction

This section describes the changes required to make test case 22.6.5 run correctly with a NBIOT UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was NBIOT testsuite which is part of iwd-TTCN3-B2013-03_D17wk37 release. This testsuite provided by MCC160 contains NBIOT test cases. 
4.2 Change 1 – Correction to function f_NBIOT_TriggerAdditionalPDN()
	Function name
	f_NBIOT_TriggerAdditionalPDN()

	Reason for change
	In this function, the CONTROL PLANE SERVICE REQ is expected to be Integrity protected and partially ciphered. This needs to be only Integrity protected (this is in line with 24.301 Sec 4.4.5).

Also, the PDN Connectivity REQ is expected to be only Integrity protected, whereas this should expected as Integrity protected and ciphered message.

This needs to be corrected.

	Summary of change
	Instead of using tsc_SHT_IntegrityProtected_PartiallyCiphered, used tsc_SHT_IntegrityProtected for capturing the CONTROL PLANNE SERVICE REQ., and used tsc_SHT_IntegrityProtected_Ciphered instead of tsc_SHT_IntegrityProtected to capture the PDN CONNECTIVITY REQ.

Please see below.

	TTCN module
	NBIOT_ESM_Testcases.ttcn

	MCC160 Comment
	Accepted


Before change
	function f_NBIOT_TriggerAdditionalPDN () 
{
    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;
    var NB_SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);
    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;
    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;
    var template DeviceProperties v_DevicePropertiesEMM := cr_DeviceProperties('D'H) ifpresent;
    var template DeviceProperties v_DevicePropertiesESM := cr_DeviceProperties('C'H) ifpresent;
    var NBIOT_SecurityParams_Type v_SecurityParams;
    var NasCount_Type v_NasCountUL;
    var NB_SRB_Identity_Type v_SrbId := f_NBIOT_GetSrb1Id(p_IOT_State);
    // Cause the UE to request connectivity to an additional PDN.
    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), p_APN_String, -, -, p_NSLPI);
    if (pc_HCCPCIoT and pc_IP_PDN){
      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;
    }
    if (isvalue(p_DeviceValue)){
      v_DevicePropertiesEMM := cr_DeviceProperties('D'H, valueof(p_DeviceValue));
      v_DevicePropertiesESM := cr_DeviceProperties('C'H, valueof(p_DeviceValue));
    }
    v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,
                                 v_PDNType,

                                 p_AccessPointName,                                                                            

                                 v_HeaderCompressConfigRX);

             v_PDNConnectivityReq_Expected.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier := p_PTI;
    v_PDNConnectivityReq_Expected.pDN_CONNECTIVITY_REQUEST.esmInfoTransferFlag := omit;
    v_PDNConnectivityReq_Expected.pDN_CONNECTIVITY_REQUEST.deviceProperties := v_DevicePropertiesESM;
    v_SecurityParams := f_NBIOT_Security_Get();
    //Step 2a1/2b1
    v_NAS_Ind := f_NBIOT_RRC_ConnEst_DefWithNas(p_CellId,
                                                p_EstablishmentCause,
                                                cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, 
                                                                   cr_CONTROL_PLANE_SERVICE_REQUEST(?, -, -, -, v_DevicePropertiesEMM)));
    if (p_IOT_State == USER_PLANE) {
      //Step 2a2
      // The SS establishes a DRB associated with the default EPS bearer context activated during the preamble.
      // Activate security in AS with parameters from AKA performed in preamble
      v_NasCountUL := v_NAS_Ind.SecurityProtection.NasCount;
      v_SecurityParams := f_NBIOT_RRC_ActivateSecurity(p_CellId, v_SecurityParams, v_NasCountUL);
      f_NBIOT_Security_Set(v_SecurityParams);
      f_NBIOT_RRCConnectionReconfiguration_SingleDRB_Est (p_CellId);
    }
    //Step 2a3/2b2
    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,
                                   v_SrbId,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                  cs_SERVICE_ACCEPT)));
    //Step 2a4/2b3
    SRB.receive (car_NB_SRB_NasPdu_IND(p_CellId,
                                         v_SrbId,
                                         cr_NAS_Indication (tsc_SHT_IntegrityProtected, 
                                                            v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp;
    return v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;
  }

	


After change
	function f_NBIOT_TriggerAdditionalPDN (NBIOT_CellId_Type p_CellId,
                                         IOT_STATE_Type p_IOT_State,
                                         ProcedureTransactionIdentifier p_PTI,
                                         template (present) AccessPointName p_AccessPointName,
                                         charstring p_APN_String,
                                         template (present) EstablishmentCause_NB_r13 p_EstablishmentCause := cr_EstablishmentCause_NB_Any,
                                         template B1_Type p_DeviceValue := omit,
                                         integer p_NSLPI := -1) runs on NBIOT_PTC return NAS_UL_Message_Type
  {
    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;
    var NB_SRB_COMMON_IND v_ReceivedAsp;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var template (present) B3_Type v_PDNType := (f_GetPdnType(), '101'B);
    var template (present) NAS_UL_Message_Type v_PDNConnectivityReq_Expected;
    var template HeaderCompressConfig v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny ifpresent;
    var template DeviceProperties v_DevicePropertiesEMM := cr_DeviceProperties('D'H) ifpresent;
    var template DeviceProperties v_DevicePropertiesESM := cr_DeviceProperties('C'H) ifpresent;
    var NBIOT_SecurityParams_Type v_SecurityParams;
    var NasCount_Type v_NasCountUL;
    var NB_SRB_Identity_Type v_SrbId := f_NBIOT_GetSrb1Id(p_IOT_State);
    // Cause the UE to request connectivity to an additional PDN.
    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), p_APN_String, -, -, p_NSLPI);
    if (pc_HCCPCIoT and pc_IP_PDN){
      v_HeaderCompressConfigRX := cr_HeaderCompressConfigAny;
    }
    if (isvalue(p_DeviceValue)){
      v_DevicePropertiesEMM := cr_DeviceProperties('D'H, valueof(p_DeviceValue));
      v_DevicePropertiesESM := cr_DeviceProperties('C'H, valueof(p_DeviceValue));
    }
v_PDNConnectivityReq_Expected := cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach,

 v_PDNType,
                                                                      p_AccessPointName,
                                                                      v_HeaderCompressConfigRX);
    v_PDNConnectivityReq_Expected.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier := p_PTI;

    v_PDNConnectivityReq_Expected.pDN_CONNECTIVITY_REQUEST.esmInfoTransferFlag := omit;
    v_PDNConnectivityReq_Expected.pDN_CONNECTIVITY_REQUEST.deviceProperties := v_DevicePropertiesESM;
    v_SecurityParams := f_NBIOT_Security_Get();
    //Step 2a1/2b1
    v_NAS_Ind := f_NBIOT_RRC_ConnEst_DefWithNas(p_CellId,
                                                p_EstablishmentCause,
                                                cr_NAS_Indication (tsc_SHT_IntegrityProtected, 
                                                                   cr_CONTROL_PLANE_SERVICE_REQUEST(?, -, -, -, v_DevicePropertiesEMM)));
    if (p_IOT_State == USER_PLANE) {
      //Step 2a2
      // The SS establishes a DRB associated with the default EPS bearer context activated during the preamble.
      // Activate security in AS with parameters from AKA performed in preamble
      v_NasCountUL := v_NAS_Ind.SecurityProtection.NasCount;
      v_SecurityParams := f_NBIOT_RRC_ActivateSecurity(p_CellId, v_SecurityParams, v_NasCountUL);
      f_NBIOT_Security_Set(v_SecurityParams);
      f_NBIOT_RRCConnectionReconfiguration_SingleDRB_Est (p_CellId);
    }
    //Step 2a3/2b2
    SRB.send(cas_NB_SRB_NasPdu_REQ(p_CellId,
                                   v_SrbId,
                                   cs_TimingInfo_Now,
                                   cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                  cs_SERVICE_ACCEPT)));
    //Step 2a4/2b3
    SRB.receive (car_NB_SRB_NasPdu_IND(p_CellId,
                                         v_SrbId,
                                         cr_NAS_Indication (tsc_SHT_IntegrityProtected_Ciphered, 
                                                            v_PDNConnectivityReq_Expected))) -> value v_ReceivedAsp;
    return v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg;
  }

	


4.3 Change 2 – Correction to function f_TC_22_6_5_NBIOT()
	Function name
	f_TC_22_6_5_NBIOT ()

	Reason for change
	In the current TTCN :

1. There is no function to prompt the user to SWITCH ON the UE after the USIM prompt. This needs to be added. Prior to this, the Ncell 1 needs to be set as serving cell as well.
2. As per the pre conditions mentioned in 36.523-1 Sec 22.6.5.3.1, the UE should be in Idle mode. In current TTCN after the registration, the rrc connection is not being released. This also needs to be implemented.

	Summary of change
	1.  Called function f_NBIOT_SetCellPower, and then called function f_UT_SwitchOnUE(UT, false);

2.  Called function f_NBIOT_RRC_ConnectionRelease() after f_NBIOT_InitialRegistration_Step5_14.

Please see below

	TTCN module
	NBIOT_ESM_Testcases.ttcn

	MCC160 Comment
	Accepted


Before change
	function f_TC_22_6_5_NBIOT () 
{ /* NB-IoT / UE requested PDN connectivity procedure not accepted / UE
     * requested PDN connectivity accepted Dual Priority T3396 override UE
     * requested PDN connectivity accepted / T3346 override */
    var IOT_STATE_Type v_IOT_State := CONTROL_PLANE;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var template (present) NAS_UL_Message_Type v_AttachReq;
    var template (present) B3_Type v_PDNTypeExpected := (f_GetPdnType(), '101'B);
    if ((pc_S1_U_DataTransfer or pc_User_Plane_CIoT_Optimisation) and pc_NB_MultiDRB) {
      v_IOT_State := USER_PLANE;
    }
    f_NBIOT_Init(c1, USER_PLANE); // even if this is run in CONTROL_PLANE test, set this flag to USER_PLANE to force the UE to configure a PDN
    // Create and configure cell
    f_NBIOT_CellConfig_Def(nbiot_Cell1, v_IOT_State); 
    //Equip UE with a USIM containing default values except for those shown in TC 22.6.5
    f_UT_USIM_Insert(UT, "36.523-1 Table 22.6.5.3.1-1");
    v_AttachReq := cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT);
    v_AttachReq.aTTACH_REQUEST.deviceProperties := cr_DeviceProperties('D'H, '1'B);
    // Receive RRC Connection Setup Complete with NAS
    v_NAS_Ind := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell1, 
                                                cr_EstablishmentCause_NB_Any,
                                                cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                  v_AttachReq,
                                                                                  cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNTypeExpected)));
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell1,v_IOT_State, STATE3_NB_IDLEUPDATED, v_NAS_Ind); 
    f_NBIOT_TestBody_Set(true);
    fl_TC_22_6_5_Body (v_IOT_State);
    f_NBIOT_TestBody_Set(false);
    f_NBIOT_Postamble(nbiot_Cell1, v_IOT_State, N1_IDLE); 
  }

	


After change
	function f_TC_22_6_5_NBIOT () 
{ /* NB-IoT / UE requested PDN connectivity procedure not accepted / UE
     * requested PDN connectivity accepted Dual Priority T3396 override UE
     * requested PDN connectivity accepted / T3346 override */
    var IOT_STATE_Type v_IOT_State := CONTROL_PLANE;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    var template (present) NAS_UL_Message_Type v_AttachReq;
    var template (present) B3_Type v_PDNTypeExpected := (f_GetPdnType(), '101'B);
    if ((pc_S1_U_DataTransfer or pc_User_Plane_CIoT_Optimisation) and pc_NB_MultiDRB) {
      v_IOT_State := USER_PLANE;
    }
    f_NBIOT_Init(c1, USER_PLANE); // even if this is run in CONTROL_PLANE test, set this flag to USER_PLANE to force the UE to configure a PDN
    // Create and configure cell
    f_NBIOT_CellConfig_Def(nbiot_Cell1, v_IOT_State); 
    //Equip UE with a USIM containing default values except for those shown in TC 22.6.5
    f_UT_USIM_Insert(UT, "36.523-1 Table 22.6.5.3.1-1");
    // First switch cell back on
    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);
    // Switch on UE
    f_UT_SwitchOnUE(UT, false);
    v_AttachReq := cr_CIOT_ATTACH_REQUEST(cr_AdditionalUpdateType_CIOT);
    v_AttachReq.aTTACH_REQUEST.deviceProperties := cr_DeviceProperties('D'H, '1'B);
    // Receive RRC Connection Setup Complete with NAS
    v_NAS_Ind := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell1,
                                                cr_EstablishmentCause_NB_Any,
                                                cr_NAS_IndicationWithPiggybacking((tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected),
                                                                                  v_AttachReq,
                                                                                  cr_CIOT_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNTypeExpected)));
    f_NBIOT_InitialRegistration_Step5_14(nbiot_Cell1, v_IOT_State, STATE3_NB_IDLEUPDATED, v_NAS_Ind); 
    //The SS transmits an RRCConnectionRelease-NB message
    f_NBIOT_RRC_ConnectionRelease(nbiot_Cell1);
    f_NBIOT_TestBody_Set(true);
    fl_TC_22_6_5_Body (v_IOT_State);
    f_NBIOT_TestBody_Set(false);
    f_NBIOT_Postamble(nbiot_Cell1, v_IOT_State, N1_IDLE);
  }

	


4.4 Change 3 – Correction to function fl_TC_22_6_5_Body()
	Function name
	fl_TC_22_6_5_Body ()

	Reason for change
	In the current TTCN :

1. Timer T3346 is being set as 5s, whereas it should be set to 300s. This is specified in Special message content for Step 16. 
2. For Steps Step 1 - 2b3/ Step 5 - 6b3/ Step 13 - 15b3/ Step 18 - 19b3/ Step 31 - 32b3, the PTI is being set to 1. This is incorrect because the PTI gets incremented as the testcase progresses. Hence, it is not advisable to hardcode this value. This needs to be modified. 

3. Before steps 3/8/16/20/33, the PTI should be extracted from the received PDN Connectivity Request and then this value should be passed as a parameter in the steps 3/8/16/20/33. This needs to be implemented.
4. At Step 26, the CONTROL PLACE SERVICE REQ is expected to be integrity protected and partially ciphered. This needs to be changed to Integrity protected, which is in line with 24.301 Sec 4.4.5. Also, before Step 26, a function to trigger the CONTROL PLANCE SERVICE REQ needs to be called. This is also missing.

	Summary of change
	1. Updated the timer T3346 to 300s

2.  Instead of setting the PTI to be tsc_PTI_1 in function f_NBIOT_TriggerAdditionalPDN , set it to ?
3.  Defined var ProcedureTransactionIdentifier v_PTId_UE and used it as shown below (Please see screenshot)
4.  Instead of using tsc_SHT_IntegrityProtected_PartiallyCiphered at Step 26, used tsc_SHT_IntegrityProtected. Also called function f_UT_RequestAdditionalPDNConnectivity before Step 26.

Please see the screenshot below.

	TTCN module
	NBIOT_ESM_Testcases.ttcn

	MCC160 Comment
	1, 4. Accepted
2,3. Accepted subject to prose CR approval


Before change
	function fl_TC_22_6_5_Body(IOT_STATE_Type p_IOT_State) runs on NBIOT_PTC
  {
    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;
    var NB_SRB_Identity_Type v_SrbId := f_NBIOT_GetSrb1Id(p_IOT_State);
    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    timer t_T3346 := f_NBIOT_SetTimerToleranceMax(nasTimer, 5.0); 
    var ProcedureTransactionIdentifier v_PTId_UE; 
    //@siclog "Step 1 - 2b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, tsc_PTI_1, cr_AccessPointName (tsc_APN1), tsc_APN1_String); 
    //@siclog "Step 3" siclog@
    SRB.send (cas_NB_SRB_NasPdu_REQ (nbiot_Cell1, 
                                     v_SrbId,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_PdnConnectReject('0'H, // no EPS bearer assigned
                                                                    tsc_PTI_1, 

                                                                    cs_ESM_Cause_v (tsc_EmmCause_ProtocolErrorUnspecified), -, -,
                                                                    omit))));
    //@siclog "Step 4" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1); 
    //@siclog "Step 5 - 6b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, '02'O, cr_AccessPointName (tsc_APN2), tsc_APN2_String);
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step 6a4/6b3");
    //@siclog "Step 8 - 12" siclog@
    fl_TC_22_6_5_ActivateAndDeactivateDefaultBearer(nbiot_Cell1, 
                                                 p_IOT_State,
                                                 v_EpsBearerId2,
                                                 '02'O, 

                                                 cr_AccessPointName (tsc_APN2),
                                                 PDN_2,
                                                 v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions,
                                                 "Step 8");
    //@siclog "Step 13 - 15b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, tsc_PTI_1, cr_AccessPointName (tsc_APN1), tsc_APN1_String, delayTolerantAccess_v1330, '1'B); 
    //@siclog "Step 16" siclog@
    SRB.send (cas_NB_SRB_NasPdu_REQ (nbiot_Cell1, 
                                     v_SrbId,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_PdnConnectReject('0'H, // no EPS bearer assigned
                                                                    tsc_PTI_1, 

                                                                    cs_ESM_Cause_v ('00011010'B),
                                                                    cs_GprsTimer3 ('37'O, tsc_Gprs3TimerUnit_1min, '00101'B), // 5mins
                                                                    -,
                                                                    omit))));
    t_T3346.start;
    //@siclog "Step 17" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1); 
    //@siclog "Step 18 - 19b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, tsc_PTI_1, cr_AccessPointName (tsc_APN1), tsc_APN1_String, -, '0'B, 1); 
    t_T3346.stop;
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step 19a4/19b3");
    //@siclog "Step 20 - 25" siclog@
    fl_TC_22_6_5_ActivateAndDeactivateDefaultBearer(nbiot_Cell1, 
                                                 p_IOT_State,
                                                 v_EpsBearerId2,
                                                 tsc_PTI_1, 

                                                 cr_AccessPointName (tsc_APN1),
                                                 PDN_1,
                                                 v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions);
    //@siclog "Step 26 - 28" siclog@
    v_NAS_Ind := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell1, 
                                                delayTolerantAccess_v1330,
                                                cr_NAS_Indication (tsc_SHT_IntegrityProtected_PartiallyCiphered, 
                                                                   cr_CONTROL_PLANE_SERVICE_REQUEST(?, -, -, -, cr_DeviceProperties('D'H, '1'B))));
    //@siclog "Step 29" siclog@
    SRB.send (cas_NB_SRB_NasPdu_REQ (nbiot_Cell1,                                     v_SrbId,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_SERVICE_REJECT(tsc_Cause_Congestion,
                                                                      -,
                                                                      cs_GPRS_Timer2_3_IEI ('5F'O, tsc_GprsTimerUnit_1min, '00101'B))))); // 5mins
    t_T3346.start;
    //@siclog "Step 30" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1); 
    //@siclog "Step 31 - 32b3" siclog@

    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, tsc_PTI_1, cr_AccessPointName (tsc_APN1), tsc_APN1_String, -, '0'B, 1); 
    t_T3346.stop;
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step 32a3/32b3");
    //@siclog "Step 33 - 38" siclog@
    fl_TC_22_6_5_ActivateAndDeactivateDefaultBearer(nbiot_Cell1, 
                                                 p_IOT_State,
                                                 v_EpsBearerId2,
                                                 tsc_PTI_1, 

                                                 cr_AccessPointName (tsc_APN1),
                                                 PDN_1,
                                                 v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions);
  }
}

	


After change
	function fl_TC_22_6_5_Body(IOT_STATE_Type p_IOT_State) runs on NBIOT_PTC
  {
    var EPS_BearerIdentity v_EpsBearerId2 := '6'H;
    var NB_SRB_Identity_Type v_SrbId := f_NBIOT_GetSrb1Id(p_IOT_State);
    var NAS_UL_Message_Type v_PDNConnectivityReq_Rxd;
    var NAS_MSG_Indication_Type v_NAS_Ind;
    timer t_T3346 := f_NBIOT_SetTimerToleranceMax(nasTimer, 300.0); 
    var ProcedureTransactionIdentifier v_PTId_UE;
    //@siclog "Step 1 - 2b3" siclog@
v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, ?, cr_AccessPointName (tsc_APN1), tsc_APN1_String); 

v_PTId_UE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    //@siclog "Step 3" siclog@
    SRB.send (cas_NB_SRB_NasPdu_REQ (nbiot_Cell1, 
                                     v_SrbId,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_PdnConnectReject('0'H, // no EPS bearer assigned
                                                                    v_PTId_UE, 
                                                                    cs_ESM_Cause_v (tsc_EmmCause_ProtocolErrorUnspecified), -, -,
                                                                    omit))));
    //@siclog "Step 4" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1); 
    //@siclog "Step 5 - 6b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, ?, cr_AccessPointName (tsc_APN2), tsc_APN2_String); 
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step 6a4/6b3");
    v_PTId_UE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    //@siclog "Step 8 - 12" siclog@
    fl_TC_22_6_5_ActivateAndDeactivateDefaultBearer(nbiot_Cell1, 
                                                 p_IOT_State,
                                                 v_EpsBearerId2,
                                                 v_PTId_UE, 
                                                 cr_AccessPointName (tsc_APN2),
                                                 PDN_2,
                                                 v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions,
                                                 "Step 8");
    //@siclog "Step 13 - 15b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, ?, cr_AccessPointName (tsc_APN1), tsc_APN1_String, delayTolerantAccess_v1330, '1'B); 

v_PTId_UE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    //@siclog "Step 16" siclog@
    SRB.send (cas_NB_SRB_NasPdu_REQ (nbiot_Cell1, 
                                     v_SrbId,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_PdnConnectReject('0'H, // no EPS bearer assigned
                                                                    v_PTId_UE, 
                                                                    cs_ESM_Cause_v ('00011010'B),
                                                                    cs_GprsTimer3 ('37'O, tsc_Gprs3TimerUnit_1min, '00101'B), // 5mins
                                                                    -,
                                                                    omit))));
    t_T3346.start;
    //@siclog "Step 17" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1); 
    //@siclog "Step 18 - 19b3" siclog@
v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, ?, cr_AccessPointName (tsc_APN1), tsc_APN1_String, -, '0'B, 1); 

    t_T3346.stop;
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step 19a4/19b3");
    v_PTId_UE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    //@siclog "Step 20 - 25" siclog@
    fl_TC_22_6_5_ActivateAndDeactivateDefaultBearer(nbiot_Cell1, 
                                                 p_IOT_State,
                                                 v_EpsBearerId2,
                                                 v_PTId_UE, 
                                                 cr_AccessPointName (tsc_APN1),
                                                 PDN_1,
                                                 v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions);
    // Cause the UE to request connectivity to an additional PDN.
    f_UT_RequestAdditionalPDNConnectivity(UT, hex2int(v_EpsBearerId2), tsc_APN1_String, -, -, -1);
    //@siclog "Step 26 - 28" siclog@
    v_NAS_Ind := f_NBIOT_RRC_ConnEst_DefWithNas(nbiot_Cell1, 
                                                delayTolerantAccess_v1330,
                                                cr_NAS_Indication (tsc_SHT_IntegrityProtected,
                                                                   cr_CONTROL_PLANE_SERVICE_REQUEST(?, -, -, -, cr_DeviceProperties('D'H, '1'B))));
    //@siclog "Step 29" siclog@
    SRB.send (cas_NB_SRB_NasPdu_REQ (nbiot_Cell1, 
                                     v_SrbId,
                                     cs_TimingInfo_Now,
                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,
                                                cs_508_SERVICE_REJECT(tsc_Cause_Congestion,
                                                                      -,
                                                                      cs_GPRS_Timer2_3_IEI ('5F'O, tsc_GprsTimerUnit_1min, '00101'B))))); // 5mins
    t_T3346.start;
    //@siclog "Step 30" siclog@
    f_NBIOT_RRC_ConnectionRelease (nbiot_Cell1); 
    //@siclog "Step 31 - 32b3" siclog@
    v_PDNConnectivityReq_Rxd := f_NBIOT_TriggerAdditionalPDN (nbiot_Cell1, p_IOT_State, ?, cr_AccessPointName (tsc_APN1), tsc_APN1_String, -, '0'B, 1); 
    t_T3346.stop;
    f_NBIOT_PreliminaryPass (__FILE__, __LINE__, "Step 32a3/32b3");
    v_PTId_UE := v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;
    //@siclog "Step 33 - 38" siclog@
    fl_TC_22_6_5_ActivateAndDeactivateDefaultBearer(nbiot_Cell1, 
                                                 p_IOT_State,
                                                 v_EpsBearerId2,
                                                 v_PTId_UE, 
                                                 cr_AccessPointName (tsc_APN1),
                                                 PDN_1,
                                                 v_PDNConnectivityReq_Rxd.pDN_CONNECTIVITY_REQUEST.extdProtocolConfigurationOptions);
  }
}

	


5 Branches executed in test case 22.6.5
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Spreadtrum RDA8909
The Spreadtrum RDA8909 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_22_6_5_RDA8909.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

